

Lecture 5: Determining Sample Size

Montgomery: Section 3.7 and 13.4

Choice of Sample Size: Fixed Effects

- Can determine the sample size for
 - Overall F test
 - Contrasts of interest
- For simplicity, typically assume n_i 's constant, i.e.,
$$n_1 = n_2 = \cdots = n_a = n$$
- Recall
 - Type I error rate: $\alpha = P(\text{Reject } H_0 | H_0)$
 - Type II error rate: $\beta = P(\text{Accept } H_0 | H_1)$
 - Power = $P(\text{Reject } H_0 | H_1) = 1 - \beta$
- Need to know
 - Test Statistics
 - Distr. of test statistics under $H_0 \implies$ Reject Region (for given α)
 - Distr. of test statistics under $H_1 \implies$ power = $P(\text{Reject Region} \mid H_1)$

Determining Power for F Test

- $\alpha = \Pr(F_0 > F_{\alpha, a-1, N-a} | H_0)$
- $\beta = \Pr(F_0 < F_{\alpha, a-1, N-a} | H_1)$
- Need to know distribution of F_0 when H_1 is true
 - Can show $F_0 = \text{MS}_{\text{Trt}} / \text{MS}_{\text{E}} \sim F_{a-1, N-a}(\delta)$
 - $\delta = n \sum \tau_i^2 / \sigma^2$ (non-centrality parameter)
- Recall $E(\text{MS}_{\text{Trt}}) = \sigma^2 + n \sum \tau_i^2 / (a - 1)$
 - $\delta = \{E(\text{MS}_{\text{Trt}}) - E(\text{MS}_{\text{E}})\} \times df_{\text{Trt}} / E(\text{MS}_{\text{E}})$
- Need to specify $\{\tau_i\}$ (Note the zero-sum constraint: $\sum_{i=1}^a \tau_i = 0$)
- Power will vary for different choices

Power Calculation for F Test

- Given α , a , and n , can determine $F_{\alpha, a-1, N-a}$
- Given some value of δ , can use noncentral F to compute power
 - In SAS, use function PROBF
 - $\text{Power} = 1 - \text{PROBF}(F_{\alpha, a-1, N-a}, a - 1, N - a, \delta)$
- Montgomery: OCC given in Chart V
 - Plots β vs Φ
 - $\Phi^2 = \delta/a = n \sum \tau_i^2 / (a\sigma^2)$
 - Can use charts to determine power or sample size

Methods to Determine δ or Φ^2

1. Choose treatment means ($\mu + \tau_i$)
 - Solve for $\{\tau_i\}$ and compute Φ^2 or δ
 - Difficult to know what means to select
2. Take a minimum difference approach
 - Suppose there exists a pair of (i, j) such that $|\tau_i - \tau_j| \geq D$
 - The minimum value: $\Phi^2 = nD^2 / (2a\sigma^2)$ (e.g., $\{\tau_i\} = \{-D/2, 0, \dots, 0, D/2\}$)
 - Power of test is at least $1 - \beta$
3. Specify a standard deviation increase in percentage (P)
 - Under H_1 , variance of a randomly chosen y_i is $\sigma_y^2 = \sigma^2 + \sum \tau_i^2 / a$
 - Randomly chosen τ_i has mean 0 and variance $\sum \tau_i^2 / a$
 - $P = \left(\sqrt{\sigma^2 + \sum \tau_i^2 / a} / \sigma - 1 \right) \times 100$
 - $\delta = an\{(1 + .01P)^2 - 1\}$
 - $\Phi^2 = n\{(1 + .01P)^2 - 1\}$

Power Calculation for Specific Contrast

- Often with an experiment, a researcher is primarily interested in just a few comparisons or contrasts. In these cases, it can be preferable to determine sample size for these rather than the overall F test.
- This reduces problem back to the t test situation
- Need to determine
 - Difference of importance
 - Standard error of comparison
- May want/need to adjust for multiple comparisons
- Montgomery describes confidence interval approach
 - Consider pairwise difference in treatment means
 - Specify length of $(1 - \alpha) \times 100\%$ confidence interval
 - $\text{Length}/2 = t_{\alpha/2, N-a} \sqrt{\frac{2\text{MS}_E}{n}}$
 - Based on the choice of MS_E , find n

Example 3.1 – Etch Rate (Page 75)

- Consider new experiment to investigate 5 RF power settings equally spaced between 180 and 200 W
- Wants to determine sample size to detect a mean difference of $D=30$ (Å/min) with 80% power
- Will use Example 3.1 estimates to determine new sample size

$$\hat{\sigma}^2 = 333.7, D = 30, \text{ and } \alpha = .05$$

- Using Table V : $\Phi^2 = 900 \times n / (2 \times 5 \times 333.7) \approx .27 \times n$

n	9	10	11
Φ	$\sqrt{2.43} \approx 1.56$	$\sqrt{2.70} \approx 1.64$	$\sqrt{3.0} \approx 1.72$
df_E	40	45	50
β	26%	20%	15%
power	74%	80%	85%

Using SAS : $\delta = a\Phi^2$

```
data new; a=5; alpha=.05; d=30; var=333.7;
  do n=5 to 15;
    df = a*(n-1);
    nc = n*d*d/(2*var);
    fcut = finv(1-alpha,a-1,df);
    beta = probf(fcut,a-1,df,nc);
    power = 1-beta;    output;
  end;
proc print;
  var n df nc beta power; run;
```

Obs	n	df	nc	beta	power
1	5	20	6.7426	0.57654	0.42346
2	6	25	8.0911	0.47884	0.52116
3	7	30	9.4396	0.39034	0.60966
4	8	35	10.7881	0.31289	0.68711
5	9	40	12.1366	0.24703	0.75297
6	10	45	13.4852	0.19234	0.80766
7	11	50	14.8337	0.14788	0.85212
8	12	55	16.1822	0.11239	0.88761
9	13	60	17.5307	0.08451	0.91549
10	14	65	18.8792	0.06292	0.93708
11	15	70	20.2277	0.04641	0.95359

***n=10 needed

- Compare all pairs and detect any difference more than 30 (Å/min) with 80% power
- A pairwise comparison takes $t_0 = (\bar{Y}_{i.} - \bar{Y}_{k.}) / \sqrt{2\text{MSE}/n}$
- Consider Tukey's adjustment: reject when $|t_0| > q_{.05}(5, df_E) / \sqrt{2}$
- $t_0 \stackrel{H_1}{\sim} t_{df_E}(\delta)$ with $\delta = (\mu_i - \mu_k) / \sqrt{2\sigma^2/n}$,
- Use PROBMC in SAS to get the quantile for multiple comparisons

```
data new1; a=5; alpha=.05; var=333.7; d=30;
  do n=8 to 12;
    df = a*(n-1); nc = d/sqrt(var*2/n);
    /* crit = tinv(1-alpha/2,df); /* LSD approach*/
    crit = probmc("range",.,1-alpha,df,a)/sqrt(2); /*Tukey*/
    power=1-probt(crit,df,nc)+probt(-crit,df,nc); output;
  end;
proc print; var n df power; run;
```

Obs	n	df	power	
1	8	35	0.65814	
2	9	40	0.73085	
3	10	45	0.79139	
4	11	50	0.84057	*** 11 replicates needed here
5	12	55	0.87971	

- Interested in comparing each setting to 200 W, and detect a difference more than 30 (Å/min) with 80% power
- A pairwise comparison takes $t_0 = (\bar{Y}_{i.} - \bar{Y}_{c.}) / \sqrt{2\text{MSE}/n}$
- Consider using Dunnett's adjustment: reject when $|t_0| > d_{.05}(4, df_E)$
- $t_0 \stackrel{H_1}{\sim} t_{df_E}(\delta)$ with $\delta = (\mu_i - \mu_k) / \sqrt{2\sigma^2/n}$,

```
data new1; a=5; alpha=.05; var=333.7; d=30;
  do n=7 to 12;
    df = a*(n-1); nc = d/sqrt(var*2/n);
    crit = probmc("dunnett2",.,1-alpha,df,a-1); /*Two-Sided Dunnett*/
    power=1-probt(crit,df,nc)+probt(-crit,df,nc);  output;
  end;
proc print;
  var n df power; run;
```

Obs	n	df	power	
1	7	30	0.68794	
2	8	35	0.76201	
3	9	40	0.82136	***Only 9 replicates needed here
4	10	45	0.86780	
5	11	50	0.90341	
6	12	55	0.93024	

Use of PROC POWER in SAS

- PROC POWER and PROC GLMPower both calculate $\sum_{i=1}^a \tau_i^2$ using pre-specified treatment means μ_i

```
proc power;
  onewayanova test=overall power=.80 npergroup=. stddev=18.27
    groupmeans = -15|0|0|0|15;
run; quit;
```

Overall F Test for One-Way ANOVA

Fixed Scenario Elements

Method	Exact
Group Means	-15 0 0 0 15
Standard Deviation	18.27
Nominal Power	0.8
Alpha	0.05

Computed N Per Group

Actual Power	N Per Group
0.808	10

- Neither PROC POWER nor PROC GLMPower can easily do multiple comparison adjustment

```
proc power;
  onewayanova test=contrast power=.80 npergroup=. stddev=18.27
    groupmeans = -15|0|0|0|15
    contrast = (1 0 0 0 -1);
run; quit;
```

Single DF Contrast in One-Way ANOVA

Fixed Scenario Elements	
Method	Exact
Contrast Coefficients	1 0 0 0 -1
Group Means	-15 0 0 0 15
Standard Deviation	18.27
Nominal Power	0.8
Number of Sides	2
Null Contrast Value	0
Alpha	0.05

Computed N Per Group	
Actual Power	N Per Group
0.844	7

Use of PROC GLMPower in SAS

```
data trtmeans; input trt resp @@; datalines;
1 -15 2 0 3 0 4 0 5 15
;

proc glmpower data=trtmeans;
  class trt;
  model resp = trt;
  contrast 'trt1 vs. trt5' trt 1 0 0 0 -1;
  power stddev=18.27 alpha=0.05 ntotal=. power=0.8;
run; quit;
```

Fixed Scenario Elements

Dependent Variable	resp
Alpha	0.05
Error Standard Deviation	18.27
Nominal Power	0.8

Computed N Total

Index	Type	Source	Test	DF	Error	DF	Actual	Power	N Total
1	Effect	trt		4		45		0.808	50
2	Contrast	trt1 vs. trt5		1		30		0.844	35

Choice of Sample Size: Random Effects

- Can use central F distribution
 - $(N - a)MS_E/\sigma^2 \sim \chi^2_{N-a}$
 - $(a - 1)MS_{Trt}/(\sigma^2 + n\sigma_\tau^2) \sim \chi^2_{a-1}$
 - Thus $F_0/\lambda^2 \sim F_{a-1, N-a}$, where $\lambda^2 = \frac{E(MS_{Trt})}{E(MS_E)} = 1 + n\sigma_\tau^2/\sigma^2$
 - Power: $P(F_0 > F_{\alpha, a-1, N-a} | \sigma_\tau^2 > 0) = P(F > F_{\alpha, a-1, N-a} / \lambda^2)$
 - Can specify ratio of σ_τ^2/σ^2
 - Can specify percentage increase $P = (\sqrt{\sigma^2 + \sigma_\tau^2}/\sigma - 1) \times 100$
- OCC given in Chart VI
 - Plots β vs λ
- Use SAS function PROBF
 - `power = 1-PROBF($F_{\alpha, a-1, N-a}/\lambda^2, a - 1, N - a$)`

Example: Batch Example on Slide 8 of Lecture 4

- Consider new experiment with 5 batches: a random effects problem
 - The variance estimate is $\sigma^2 = 1.8$
 - Desire to detect situation when $\sigma_\tau^2 \geq 3.6 = 2\sigma^2$
 - Set power at 80% and $\alpha = .05$
- Using Table VI : $\lambda = \sqrt{1 + 2n}$

n	3	4	5
λ	$\sqrt{7} \approx 2.65$	$\sqrt{9} = 3$	$\sqrt{11} \approx 3.32$
df_E	10	15	20
β	28%	18%	15%
power	72%	82%	85%

- Appears $n = 4$ gives appropriate power

Using SAS

```
data new; a = 5; alpha=.05; ratiovar=2.0;
  do n=2 to 10;
    df = a*(n-1);
    lambdasq = 1+ratiovar*n;
    fcut  = finv(1-alpha,a-1,df);
    beta=probf(fcut/lambdasq,a-1,df);
    power = 1-beta;    output;
  end;

proc print;
  var n beta power; run;
```

Obs	n	beta	power	
1	2	0.52933	0.47067	
2	3	0.26112	0.73888	
3	4	0.15292	0.84708	** n=4 gives the power
4	5	0.10027	0.89973	
5	6	0.07081	0.92919	
6	7	0.05267	0.94733	
7	8	0.04072	0.95928	
8	9	0.03242	0.96758	
9	10	0.02643	0.97357	